SINGULAR CODE FOR "THE TATE CONJECTURE FOR A FAMILY OF SURFACES OF GENERAL TYPE WITH $p_g = q = 1$ AND $K^2 = 3$"

CHRISTOPHER LYONS

Abstract. We describe the calculation of certain local properties of the pencil J_1 in [3], including the code for implementing these calculations in the software package Singular.

In what follows, we freely make use of the notions and notations from [3], specifically from the first three sections.

Coordinate charts

Recall that one has the equations of Ishida [2] for a basis of the sections $H^0(\tilde{P}, \Phi^* L)^G$:

$$\begin{aligned}
\Psi_1 &:= fZ_0^4 + gZ_1^4 + hZ_2^4, \\
\Psi_2 &:= Z_0Z_1Z_2(Z_0 + Z_1 + Z_2), \\
\Psi_3 &:= fZ_0^3Z_2 + gZ_1^3Z_0 + hZ_2^3Z_1, \\
\Psi_4 &:= fZ_0^3Z_1 + gZ_1^3Z_2 + hZ_2^3Z_0, \\
\Psi_5 &:= ghZ_1^2Z_2^2 + fhZ_0^2Z_2^2 + fgZ_0^2Z_1^2.
\end{aligned}$$

These equations are easily adapted to describe the sections Ψ_i away from the fibers of $\tilde{P}$ over $\{\tilde{0}, C_1, C_2\}$. We will describe coordinates on this open subset of $\tilde{P}$ and then adapt the Ψ_i to these coordinates.

Let $U := \tilde{E} \setminus \{\tilde{0}, C_1, C_2\}$. Since $y^2 = w(x)$ is an affine Weierstrass equation for $\tilde{E}$, one has

$$\tilde{E} \setminus \{\tilde{0}\} \simeq \operatorname{Spec}\left(\frac{\mathbb{C}[x,y]}{\langle y^2 - w(x)\rangle}\right)$$

and, since $C_1 = (\alpha, \beta)$ and $C_2 = (\alpha, -\beta)$, it follows that

$$U \simeq \operatorname{Spec}\left(\frac{\mathbb{C}[x,y,t]}{\langle y^2 - w(x), (x-\alpha)t - 1\rangle}\right).$$

Since the sections Z_0, Z_1, Z_2 are defined naturally to correspond to the natural basis of the decomposition

$$H^0(\tilde{P}, \mathcal{O}_{\tilde{P}}(1)) \simeq H^0(\tilde{E}, \mathcal{O}_{\tilde{E}}(\tilde{0})) \oplus H^0(\tilde{E}, \mathcal{O}_{\tilde{E}}(C_1)) \oplus H^0(\tilde{E}, \mathcal{O}_{\tilde{E}}(C_2))$$

one obtains a trivialization $\tilde{P}|_U \xrightarrow{\sim} U \times \mathbb{P}^2$ by using the relative homogenous coordinates $(Z_0 : Z_1 : Z_2)$.

Note that as the action of G on $\tilde{P}$ permutes the fibers over $\{\tilde{0}, C_1, C_2\}$, G also acts on $\tilde{P}|_U$. On $\tilde{P}|_U$ this action is given on $(Q, (Z_0 : Z_1 : Z_2)) \in \tilde{E} \times \mathbb{P}^2 \simeq \tilde{P}|_U$ by

$$\tilde{\tau}_{C_1}^*(Q \oplus C_1, (Z_2 : Z_0 : Z_1)), \qquad \tilde{\tau}_{C_2}^*(Q \oplus C_2, (Z_1 : Z_2 : Z_0)).$$

Thus G permutes the three open affines

$$\tilde{P}|_U \cap \{Z_0 \neq 0\}, \qquad \tilde{P}|_U \cap \{Z_1 \neq 0\}, \qquad \tilde{P}|_U \cap \{Z_2 \neq 0\}.$$

In particular, when studying the local properties of a G-invariant divisor on $\tilde{P}$, one can ascertain its behavior on all of $\tilde{P}|_U$ by only restricting attention to one of these three affine opens.

With this in mind, we define

$$T := \tilde{P}|_U \cap \{Z_0 \neq 0\}$$

and set $u := Z_1/Z_0, v := Z_2/Z_0$. Then one has

$$T \simeq \operatorname{Spec}\left(\frac{\mathbb{C}[x,y,u,v,t]}{\langle y^2 - w(x), (x-\alpha)t - 1\rangle}\right).$$

We will work with these coordinates to establish local properties of the sections Ψ_i in $\tilde{P}|_U$. In doing so, we prefer to work with polynomials in $\mathbb{C}[x,y,u,v,t]$ rather than the original equations Ψ_i. Upon setting $(Z_0 : Z_1 : Z_2) = (1 : u : v)$ in the equation Ψ_i, one obtains a rational function in $\mathbb{C}(x,y,u,v,t)$. By clearing denominators, one obtains a polynomial basis for $H^0(\tilde{P}, \Phi^*L)^G$ on T. More specifically, if we set

$$\begin{aligned} b_1 &:= 2\beta(y-\beta) - \mu(x-\alpha) \\ b_2 &:= -2\beta(y+\beta) - \mu(x-\alpha), \end{aligned}$$

then $f = x-\alpha$, $g = -4\beta^2(x-\alpha)/b_1$, and $h = -4\beta^2(x-\alpha)/b_2$. Furthermore, one can check that $fgh = -4\beta^2$, which yields $b_1 b_2 = -4\beta^2(x-\alpha)^3$. One sees that multiplying each of the Ψ_i by b_1b_2 will clear their common denominator, and (upon removing a common factor of $(x-\alpha)$) this yields the following choice of equations ω_i:

$$\begin{aligned} \omega_1 &:= b_1b_2 + 4\beta^2 b_2 u^4 + 4\beta^2 b_1 v^4, \\ \omega_2 &:= -4\beta^2(x-\alpha)^2 uv(1+u+v), \\ \omega_3 &:= b_1b_2 v + 4\beta^2 b_2 u^3 + 4\beta^2 b_1 v^3 u, \\ \omega_4 &:= b_1b_2 u + 4\beta^2 b_2 u^3 v + 4\beta^2 b_1 v^3, \\ \omega_5 &:= 4\beta^2(x-\alpha)\left(4\beta^2u^2v^2 + b_1v^2 + b_2u^2\right). \end{aligned}$$

The equations ω_i give a basis of $\Gamma(T, \Phi^*L)$ satisfying

$$(\omega_1 : \cdots : \omega_5) = (\Psi_1 : \cdots : \Psi_5)$$

on T.

To finish this section, recall that the elements of $H^0(\tilde{P}, \Phi^*L)^G$ near the fiber $\tilde{p}^{-1}(\tilde{0})$ are handled via the local equations

$$\chi_i\big(t, (Z_0' : Z_1 : Z_2)\big) := t^{-1}\Psi_i(tZ_0' : Z_1 : Z_2)$$

in order to account for the fact that all sections vanish to order at least one on $\tilde{p}^{-1}(\tilde{0})$. Here $t = x/y$ is a parameter at $\tilde{0}$ and $Z_0' := t^{-1}Z_0$. When f, g, h are expanded in powers of t (for details, see [2, p.39]), these equations become

$$\begin{aligned} \chi_1 &= 2\beta(Z_1^4 - Z_2^4) + t(Z_0'^4 + \mu Z_1^4 + \mu Z_2^4) + \text{(higher terms)}, \\ \chi_2 &= Z_0' Z_1 Z_2(Z_1 + Z_2) + tZ_0^2 Z_1 Z_2, \\ \chi_3 &= Z_0'^3 Z_2 - 2\beta Z_1 Z_2^3 + t(\mu Z_1 Z_2^3 + 2\beta Z_0' Z_1^3) + \text{(higher terms)}, \\ \chi_4 &= Z_0'^3 Z_1 + 2\beta Z_1^3 Z_2 + t(\mu Z_1^3 Z_2 - 2\beta Z_0' Z_2^3) + \text{(higher terms)}, \\ \chi_5 &= 2\beta Z_0'^2(Z_1^2 - Z_2^2) + t(\mu Z_0'^2 Z_2^2 + \mu Z_0'^2 Z_1^2 - 4\beta^2 Z_1^2 Z_2^2) + \text{(higher terms)}. \end{aligned}$$

These are the equations we utilize in our study of $H^0(\tilde{P}, \Phi^*L)^G$ near $\tilde{p}^{-1}(\tilde{0})$. Note that just outside $\tilde{p}^{-1}(\tilde{0})$ we have

$$(\chi_1 : \cdots : \chi_5) = (\Psi_1 : \cdots : \Psi_5).$$

As G permutes the fibers over $\{\tilde{0}, C_1, C_2\}$, studying the local behavior of a section of $H^0(\tilde{P}, \Phi^*L)^G$ near $\tilde{p}^{-1}(\tilde{0})$ will yield its behavior near the other two fibers as well. Thus we obtain a complete picture of the these sections on all of $\tilde{P}$ just by studying them on $T \cup \tilde{p}^{-1}(\tilde{0})$.

Calculations

With $J_1 \subseteq |\mathfrak{D}|$ on $E_1^{(3)}$, let $\tilde{J}_1 = \Phi^* J_1 \subseteq \Phi^*|\mathfrak{D}|$ denote the pullback of the pencil to $\tilde{P}$ via the diagram

$$\begin{array}{ccc} \tilde{P} & \xrightarrow{\Phi} & E_1^{(3)} \\ \big\downarrow{\scriptstyle \tilde{p}} & & \big\downarrow{\scriptstyle \mathrm{AJ}} \\ \tilde{E}_1 & \xrightarrow{\varphi} & E_1 \end{array}\ .$$

If A_1 is the base locus of J_1, then $\Phi^{-1}(A_1) \subseteq \tilde{P}$ is the base locus of $\tilde{J}_1$.

(1) This calculation shows that $\Phi^{-1}(A_1)$ is nonsingular in the affine coordinate chart $T \subseteq \tilde{P}$.

```
> ring R=0,(x,y,u,v,t),dp;
> poly E1_tilde=y2-x3-x2-x+3/4;
> poly alpha=1;
> poly beta=3/2;
> poly mu=6;
> poly b1=2*beta*(y-beta)-mu*(x-alpha);
> poly b2=-2*beta*(y+beta)-mu*(x-alpha);
> poly invert_t=(x-alpha)*t-1;
> poly omega_1=b1*b2+4*beta^2*b2*u4+4*beta^2*b1*v4;
> poly omega_2=-4*beta^2*(x-alpha)^2*uv*(1+u+v);
> poly omega_3=b1*b2*v+4*beta^2*b2*u3+4*beta^2*b1*v3u;
> poly omega_4=b1*b2*u+4*beta^2*b2*u3v+4*beta^2*b1*v3;
> poly omega_5=4*beta^2*(x-alpha)*(4*beta^2*u2v2+b1*v2+b2*u2);
> ideal I=E1_tilde,invert_t,omega_1,omega_3-omega_4;
> matrix J=jacob(I);
> ideal K=minor(J,4); K=K+I; K=stdfglm(K);
> dim(K);
-1 // No solutions.
```

(2) This calculation shows that $\Phi^{-1}(A_1)$ has no singularities in the fiber $\tilde{p}^{-1}(\tilde{0}) \subseteq \tilde{P}$.

```
> ring R=0,(u,v,w,t),dp;
> poly beta=3/2;
> poly mu=6;
> poly chi_1=2*beta*(v4-w4)+t*(u4+mu*v4+mu*w4);
> poly chi_2=uvw*(v+w)+tu2vw;
> poly chi_3=u3w-2*beta*vw3+t*(mu*vw3+2*beta*uv3);
> poly chi_4=u3v+8v3w+t*(mu*v3w-2*beta*uw3);
> poly chi_5=2*beta*u2*(v2-w2)+t*(mu*u2*(v2+w2)-4*beta^2*v2w2);
> poly p1=subst(chi_1,w,1);
> poly p2=subst(chi_3-chi_4,w,1);
> ideal Jac1=diff(p1,u)*diff(p2,v)-diff(p1,v)*diff(p2,u),
diff(p1,u)*diff(p2,t)-diff(p1,t)*diff(p2,u),
diff(p1,v)*diff(p2,t)-diff(p1,t)*diff(p2,v);
> ideal sing1=p1,p2,w-1,t,Jac1;
> ideal K1=std(sing1);
> dim(K1);
-1                        // No solutions with Z_0' != 0.
> poly q1=subst(chi_1,v,1);
> poly q2=subst(chi_3-chi_4,v,1);
> ideal Jac2=diff(q1,u)*diff(q2,w)-diff(q1,w)*diff(q2,u),
diff(q1,u)*diff(q2,t)-diff(q1,t)*diff(q2,u),
diff(q1,w)*diff(q2,t)-diff(q1,t)*diff(q2,w);
> ideal sing2=q1,q2,v-1,t,Jac2;
> ideal K2=std(sing2);
> dim(K2);
-1                         // No solutions with Z_1 != 0.
> poly r1=subst(chi_1,u,1);
> poly r2=subst(chi_3-chi_4,u,1);
> ideal Jac3=diff(r1,w)*diff(r2,v)-diff(r1,v)*diff(r2,w),
diff(r1,w)*diff(r2,t)-diff(r1,t)*diff(r2,w),
diff(r1,v)*diff(r2,t)-diff(r1,t)*diff(r2,v)
> ideal sing3=r1,r2,u-1,t,Jac3;
> ideal K3=std(sing3);
```

```
> dim(K3);
-1                          // No solutions with Z_2 != 0.
```

(3) This calculation finds all fibers of $\tilde{J}_1$ that have singularities in T, and gives numerical coordinates for them. The numerical coordinates are contained in the separate file [4].

```
> ring R=0,(x,y,u,v,t,a),(dp(5),dp(1));
> poly E1_tilde=y2-x3-x2-x+3/4;
> poly alpha=1;
> poly beta=3/2;
> poly mu=6;
> poly b_1=2*beta*(y-beta)-mu*(x-alpha);
> poly b_2=-2*beta*(y+beta)-mu*(x-alpha);
> poly invert_t=(x-alpha)*t-1;
> poly omega_1=b_1*b_2+4*beta^2*b_2*u4+4*beta^2*b_1*v4;
> poly omega_2=-4*beta^2*(x-alpha)^2*uv*(1+u+v);
> poly omega_3=b_1*b_2*v+4*beta^2*b_2*u3+4*beta^2*b_1*v3u;
> poly omega_4=b_1*b_2*u+4*beta^2*b_2*u3v+4*beta^2*b_1*v3;
> poly omega_5=4*beta^2*(x-alpha)*(4*beta^2*u2v2+b_1*v2+b_2*u2);
> poly L1=omega_1+a*(omega_3-omega_4);
> ideal sing=E1_tilde,L1,invert_t,diff(L1,u),diff(L1,v),
diff(E1_tilde,y)*diff(L1,x)-diff(E1_tilde,x)*diff(L1,y);
> ideal K=stdfglm(sing); dim(K); vdim(K);
0                  // Solution set has dimension 0 as affine scheme
126                // Solution set has dimension 126 as vector space over Rationals
> poly sing_a_values=K[1];
                   // Given the choice of monomial ordering above, the first
                    element in a Groebner basis for the ideal "sing" will represent
                    the intersection of "sing" with the polynomial subring in the
                    variable a.
> sing_a_values;
751689a42+133446906a40+12912311529a38+517896020340a36
+7891332449328a34+55050945738624a32+224530526292224a30
+594950582418432a28+1110418441469952a26+1912067423830016a24
+4686348602572800a22+12979649665302528a20
+28031314997280768a18+43741902922579968a16
+49919629706919936a14+42303198374920192a12
+26690144132136960a10+12381904309321728a8
+4097476109795328a6+913333385428992a4
+122458107543552a2+7421703487488
                    // This shows there are 42 values of a for which
                      the fiber is singular.  Note also that a=0 is
                      not a root since the constant term is nonzero.
> LIB "solve.lib"; solve(K,15,1);
                    // This command gives the numerical coordinates for the 126
                      solutions.  The output from this command has been placed
                      in the separate file "SING_AWAY_FROM_0.pdf"
```

(4) This calculation shows that no fiber of $\tilde{J}_1$ has singularities in the fiber $\tilde{p}^{-1}(\tilde{0}) \subseteq \tilde{P}$.

```
> ring R=0,(u,v,w,t,a),dp;
// Here we have set u=Z_0', v=Z_1, w=Z_2.
> poly beta=3/2;
```

```
> poly mu=6;
> poly chi_1=2*beta*(v4-w4)+t*(u4+mu*v4+mu*w4);
> poly chi_2=uvw*(v+w)+tu2vw;
> poly chi_3=u3w-2*beta*vw3+t*(mu*vw3+2*beta*uv3);
> poly chi_4=u3v+8v3w+t*(mu*v3w-2*beta*uw3);
> poly chi_5=2*beta*u2*(v2-w2)+t*(mu*u2*(v2+w2)-4*beta^2*v2w2);
> poly L1=a*chi_1+(chi_3-chi_4);
> ideal sing1=L1,diff(L1,t),diff(L1,v),diff(L1,w),u-1,t;
> dim(std(sing1));
-1 // No solutions with Z_0' != 0.
> ideal sing2=L1,diff(L1,t),diff(L1,u),diff(L1,w),v-1,t;
> dim(std(sing2));
-1 // No solutions with Z_1 != 0.
> ideal sing3=L1,diff(L1,t),diff(L1,u),diff(L1,v),w-1,t;
> dim(std(sing3));
-1 // No solutions with Z_2 != 0.
```

References

[1] G.-M. Greuel, G. Pfister, H. Schönemann. SINGULAR, A System for Polynomial Computations, version 3.1. Available at `http://www.singular.uni-kl.de/`

[2] H. Ishida. Catanese-Ciliberto surfaces of fiber genus three with unique singular fiber. *Tohoku Math. J. (2)* **58** (2006), 33–69.

[3] C. Lyons. *The Tate Conjecture for a family of surfaces of general type with $p_g = q = 1$ and $K^2 = 3$.* Available at `http://mathfaculty.fullerton.edu/clyons/data/index.htm`

[4] C. Lyons. *Numerical data for "The Tate Conjecture for a family of surfaces of general type with $p_g = q = 1$ and $K^2 = 3$".* Available at `http://mathfaculty.fullerton.edu/clyons/data/index.htm`